


Exercise 2 – Explosion in air-filled rigid tank

This is a purely fluid problem because the tank is rigid and therefore does not need to be modeled. There is only one type of fluid (single-phase, single-component). This considerably simplifies the modeling, since the bubble surface may be treated as ALE.

- Try out Lagrangian solution
- Try out Eulerian solution
- Try out ALE solution
- ...



36

Geometric data:

The calculation is 2-D plane strain. The tank is 10 units wide and 12 units high. The explosive bubble is 2 units wide and 2 units high.

Materials

The explosive bubble is a perfect gas with $\gamma = 1.4$, initial density 10 and initial specific energy 2.5×10^5 , thus resulting in an initial pressure of:

$$p_0^{\text{bubble}} = (1.4 - 1) \cdot 10 \cdot 2.5 \times 10^5 = 1.0 \times 10^6$$

The low-pressure surrounding gas has the same nature but an initial density of 1, thus resulting in an initial pressure of:

$$p_0^{\text{gas}} = (1.4 - 1) \cdot 1 \cdot 2.5 \times 10^5 = 1.0 \times 10^5$$

The tank is rigid and does not need to be modelled.

Thanks to symmetries with respect to the x- and y-axes, only $\frac{1}{4}$ of the geometry needs to be modelled.

We want to study the effects of the explosion up to 10 ms of physical time.

Because the two fluids have the same nature (but different initial conditions), they may be freely intermixed with each other during the simulation. Therefore the bubble surface does not need to be modelled as Lagrangian, even by using a single-component model for the fluid.

TANK01

Lagrangian solution: the whole mesh is Lagrangian. The mesh generation file is:

```
*size 100
opti echo 1;
opti dime 2 elem qua4;
p0 = 0 0;
p1 = 5 0;
p2 = 5 6;
p3 = 0 6;
p4 = 1 0;
p5 = 1 1;
p6 = 0 1;
p7 = 5 1;
p8 = 1 6;
tol = 0.01;
n1 = 2;
n2 = 8;
n3 = 10;
c1 = p0 d n1 p4;
c2 = p4 d n1 p5;
c3 = p5 d n1 p6;
c4 = p6 d n1 p0;
bull = (dall c1 c2 c3 c4 plan) coul roug;
c1 = p4 d n2 p1;
c2 = p1 d n1 p7;
c3 = p7 d n2 p5;
c4 = p5 d n1 p4;
gas1 = (dall c1 c2 c3 c4 plan) coul turq;
c1 = p5 d n2 p7;
c2 = p7 d n3 p2;
c3 = p2 d n2 p8;
c4 = p8 d n3 p5;
gas2 = (dall c1 c2 c3 c4 plan) coul turq;
c1 = p6 d n1 p5;
c2 = p5 d n3 p8;
c3 = p8 d n1 p3;
c4 = p3 d n3 p6;
gas3 = (dall c1 c2 c3 c4 plan) coul turq;
gas = (gas1 et gas2 et gas3) coul turq;
flui = bull et gas;
elim toi flui;
symx = flui poin droi p0 p3 tol;
blox = flui poin droi p1 p2 tol;
symy = flui poin droi p0 p1 tol;
bloy = flui poin droi p3 p2 tol;
e1 = bull elem cont p0;
e2 = gas elem cont p2;
mesh = flui et symx et blox et symy et bloy et e1 et e2;
tass mesh;
opti sauv form 'tank01.msh';
sauv form mesh;
opti trac psc;
trac qual mesh;
liat (nbcl flui);
liat (nbno flui);
fin;
```

The input file is:

```
TANK - 01
*-----
*ECHO
*CONV win
CAST mesh
*-----Problem type
DEPLA NONL LAGR
*-----Dimensioning
DIME
PTZL 143 FL24 145
TERM
*-----Geometry
GEOM FL24 flui TERM
*-----Material data
MATE FLUT RO 10 EINT 2.5E5 GAMM 1.4 PB 0 ITER 1 ALFO 1 BETO 1 KINT 1
AHGF 0 CL 0.5 CQ 2.56 PMIN 0 NUM 1 LECT bull TERM
FLUT RO 1 EINT 2.5E5 GAMM 1.4 PB 0 ITER 1 ALFO 1 BETO 1 KINT 1
AHGF 0 CL 0.5 CQ 2.56 PMIN 0 NUM 1 LECT gas TERM
*-----Boundary conditions
LINK COMP
BLOO 1 LECT blox TERM
BLOO 2 LECT bloy TERM
CONT SPLA NX 1 NY 0 LECT symx TERM
CONT SPLA NX 1 NY 1 LECT symy TERM
*-----Outputs
ECRI COOR DEPL VITE CONT ECRO TPPE 1.E-3
FICH ALIC TEMP FREQ 1
POIN LECT p4 p6 TERM
ELEM LECT e1 e2 TERM
*-----Options
OPTI NOTE
CSTA 0.8
LOG 1
*-----Transient calculation
CALCUL TINI 0 TEND 10.E-3
*-----Direct Animation
PLAY
CAME 1 EYE 0.00000E+00 0.00000E+00 3.90512E+01
! Q 1.00000E+00 0.00000E+00 0.00000E+00 0.00000E+00
VIEW 0.00000E+00 0.00000E+00 -1.00000E+00
RIGH 1.00000E+00 0.00000E+00 0.00000E+00
UP 0.00000E+00 1.00000E+00 0.00000E+00
FOV 2.48819E+01


scen geom navi free
iso fill fiel ecro 1 scal user prog 0.e0 pas 0.25e5 3.25e5 term
text isca
colo pape

freq 1
sler cam1 1 nfra 1
trac offs fich avi nocl nfto 120 fps 10 kfre 10 comp -1
symx symy rend
gotr loop 118 offs fich avi cont nocl
symx symy rend
go
trac offs fich avi cont
symx symy rend


ENDPLAY
*=====POST-TREATMENT
SUIT
Post-treatment
ECHO
RESU ALIC TEMP GARD PSCR
SORT GRAP
AXTE 1000.0 'Time [ms]'
*-----Curve definitions
COUR 1 'dx_4' DEPL COMP 1 NOEU LECT p4 TERM
COUR 2 'dy_6' DEPL COMP 2 NOEU LECT p6 TERM
COUR 3 'pr_1' ECRO COMP 1 ELEM LECT e1 TERM
COUR 4 'pr_2' ECRO COMP 1 ELEM LECT e2 TERM
*-----Plots
trac 1 axes 1.0 'DISPL. [M]'
trac 2 axes 1.0 'DISPL. [M]'
trac 3 axes 1.0 'PRESS. [PA]'
trac 4 axes 1.0 'PRESS. [PA]'
*-----Results qualification
QUAL DEPL COMP 1 LECT p4 TERM REFE 2.21466E+00 TOLE 1.E-2
DEPL COMP 2 LECT p6 TERM REFE 2.21638E+00 TOLE 1.E-2
CONT COMP 1 LECT e1 TERM REFE 2.47193E+04 TOLE 1.E-2
CONT COMP 1 LECT e2 TERM REFE 1.10457E+05 TOLE 1.E-2
*=====
FIN
```

The numerical solution is able to proceed up to 10 ms. However, large distortions in the fluid mesh occur due to the large growth of the bubble. The final mesh and pressure distributions are:


TANK - 01
Time: 1.00000E-02 Step: 119



TANK - 01
Time: 1.00000E-02 Step: 119




The computed fluid pressures at the bubble center and at the corners are shown hereafter:




TANK02

Eulerian solution: the whole mesh is Eulerian. This solution is possible in the present example because there are no structures nor any Lagrangian fluid-fluid interfaces. The mesh stays constant, of course, and the final pressure distribution is


TANK - 02
Time: 1.00000E-02 Step: 31



TANK - 02
Time: 1.00000E-02 Step: 31




The computed pressure histories are:



TANK03

ALE solution: the whole mesh is ALE, except the bubble surface, that is treated as Lagrangian (note that this is not strictly necessary in the present example, see the previous Eulerian solution). The final mesh is (compare with Lagrangian case) and the final pressure distribution:



The computed pressure histories are:

